UCRL-52888
Distribution Category UC-66b

GL03929

Effect of permeability on cooling of a magmatic intrusion in a geothermal reservoir

K. H. Lau*

Manuscript date: January 11, 1980

*University of Hawaii at Hilo, LLNL Summer Institute Employee.

LAWRENCE LIVERMORE LABORATORY
University of California • Livermore, California • 94550

Available from: National Technical Information Service • U.S. Department of Commerce
5285 Port Royal Road • Springfield, VA 22161 • $6.00 per copy • (Microfiche $3.50)

CONTENTS

NOMENCLATURE

u', v'	velocity components in the x and y directions, respectively
u, v	dimensionless velocity component in the x and y directions, respectively
g	gravitational acceleration
H	depth of the reservoir floor
L	width of the reservoir
λ_m	thermal conductivity of the porous medium
λ_{cap}	thermal conductivity of the cap rock
ψ'	stream function
ψ	dimensionless stream function
θ'	temperature referenced to T_0 $(= T - T_0)$
θ	dimensionless temperature
ΔT	maximum temperature referenced to T_0 $(= T_{max} - T_0)$
x', y'	Cartesian coordinates
x, y	dimensionless coordinates
$(\rho c)_m$	heat capacity of porous medium
$(\rho c)_f$	heat capacity of fluid
Ra	Rayleigh number
K_x	permeability in x direction
K_y	permeability in y direction
μ	viscosity of fluid
ρ	density of fluid
β	thermal expansion coefficient
α_m	thermal diffusivity of fluid porous medium
t'	time
t	dimensionless time
χ	permeability ratio K_y/K_x
γ	heat capacity ratio $(\rho c)_f/(\rho c)_m$
η	dimensionless measurement from reservoir floor to cap rock
ρ_0	density of fluid at $T = T_0$
Q	surface heat flow
HFU	heat flow unit

Subscripts

f	fluid
m	country rock
i	index in x direction
j	index in y direction

Superscript

k	index in arbitrary time step k
2n + 1	index in $(2n+1)^{th}$ time step

EFFECT OF PERMEABILITY ON COOLING OF A MAGMATIC INTRUSION IN A GEOTHERMAL RESERVOIR

ABSTRACT

This report describes numerical modeling of the transient cooling of a magmatic intrusion in a geothermal reservoir that results from conduction and convection, considering the effects of overlying cap rock and differing horizontal and vertical permeabilities of the reservoir. These results are compared with data from Salton Sea Geothermal Field (SSGF). Multiple layers of convection cells are observed when horizontal permeability is much larger than vertical permeability. The sharp drop-off of surface heat flow experimentally observed at SSGF is consistent with the numerical results. We estimate the age of the intrusive body at SSGF to be between 6000 and 20,000 years.

INTRODUCTION

Because hydrothermal systems of a particular geothermal field are important in all aspects of geothermal power production, geophysicists and geothermal reservoir engineers are greatly interested in magmatic intrusions in the earth's crust. These intrusions, also known as plutons, are cooled by surrounding country rock. If the neighboring formations are permeable and saturated with ground water, then convective hydrothermal systems can result. The nature of these hydrothermal systems is determined by the physical properties of the surrounding formations.

Intrusive magma can take different forms or sizes. A sheet-like intrusive body--perpendicular to the stratification in the bedded rocks--is called a dike. Jaeger[1] and Horai[2] studied dike intrusion based on heat conduction alone. Recent studies[3-5] suggest that convection of ground water also plays an important role in heat transfer in geothermal fields.

Numerical modeling studies of dike-induced convection flow include the work of Lau and Cheng[3] on the effects of dike intrusion on steady-state temperature distribution, streamlines, and shape of water table in a volcanic island aquifer. Norton and Knight[4] researched the time dependence of convective circulation and its influence on the cooling rate of massive plutons. Torrance and Sheu[5] studied the cooling of a pluton by assuming that the intrusion itself becomes permeable below a specified thermal stress-cracking temperature.

In all of these referenced studies, the permeability is assumed constant, and the existence of cap rock is not included in the analysis. Kasameyer and Younker[6] suggested that the cap rock and a large horizontal-to-vertical permeability ratio can be responsible for the dramatic reduction in geothermal gradient in the Salton Sea Geothermal Field (SSGF).

The present study of the cooling of a magmatic intrusion because of natural convection takes into account the effects of overlying cap rock of various thicknesses as well as of differing horizontal and vertical permeabilities in the reservoir. Results are specifically related to the SSGF. Figure 1 shows an idealized model.



FIG. 1. Idealized model of a geothermal reservoir with dike intrusion.

DESCRIPTION OF MODELING PROCESS

GOVERNING EQUATIONS

The governing equations for the hydrothermal system in a porous medium are the continuity equation, Darcy's law, the energy equation, and the equation of state. With the Boussinesq approximation, these equations can be written as

$$\frac{\partial u'}{\partial x'} + \frac{\partial v'}{\partial y'} = 0 \quad , \tag{1}$$

$$u' = \left(\frac{-K_x}{\mu}\right)\left(\frac{\partial p'}{\partial x'}\right) , \tag{2}$$

$$v' = \frac{-K_y}{\mu}\left(\frac{\partial p'}{\partial y'} + \rho g\right) , \tag{3}$$

$$(\rho c)_m \frac{\partial \theta'}{\partial t'} + (\rho c)_f \left(u' \frac{\partial \theta'}{\partial x'} + v' \frac{\partial \theta'}{\partial y'}\right) = \left(\lambda_m \quad \frac{\partial^2 \theta'}{\partial x'^2} + \frac{\partial^2 \theta'}{\partial y'^2}\right) , \tag{4}$$

$$\rho = \rho_0 \ (1 - \beta\theta') \quad . \tag{5}$$

When one introduces the stream function ψ' and the following dimensionless variables,

$$u' = \frac{\partial \psi'}{\partial y'} \quad , \tag{6}$$

$$v' = \frac{\partial \psi'}{\partial x'} \quad , \tag{7}$$

$$t = \frac{\alpha_m}{H^2} t' \quad , \tag{8}$$

$$x = \frac{x'}{H} \quad , \tag{9}$$

$$y = \frac{y'}{H} \quad , \tag{10}$$

$$\theta = \frac{\theta'}{\Delta T} \quad , \tag{11}$$

$$u = \frac{u'H}{\alpha_m} \quad , \tag{12}$$

$$v = \frac{v'H}{\alpha_m} \quad , \tag{13}$$

$$\psi = \frac{\psi'}{\alpha_m} \quad , \tag{14}$$

$$Ra = \frac{\rho_0 \ \beta g \ K_y \ H \ \Delta T}{\mu \alpha_m} \quad , \tag{15}$$

the nondimensional form of the governing equations becomes

$$\frac{\partial \theta}{\partial t} + \gamma \left(\frac{\partial \psi}{\partial y}\right) \left(\frac{\partial \theta}{\partial x}\right) - \gamma \left(\frac{\partial \psi}{\partial x}\right) \left(\frac{\partial \theta}{\partial y}\right) = \frac{\partial^2 \theta}{\partial x^2} + \frac{\partial^2 \theta}{\partial y^2} \quad , \tag{16}$$

$$\frac{\partial^2 \psi}{\partial x^2} + \chi \frac{\partial^2 \psi}{\partial y^2} = -Ra \frac{\partial \theta}{\partial x} . \tag{17}$$

BOUNDARY AND INITIAL CONDITIONS

The initial conditions of the problem are $\psi = 0$ and $\theta = 0$ everywhere in the region except in the intrusive area, where $\theta = 1$. The boundary condition at the surface is a constant temperature; i.e.,

$$\theta(x,1) = 0 \quad . \tag{18}$$

The boundaries at $x = 0$ and L/H are impermeable to flow and thermally nonconductive; i.e.,

$$\frac{\partial \theta}{\partial x} (0,y) = \frac{\partial \theta}{\partial x} \left(\frac{L}{H}, y\right) = 0 \quad , \tag{19}$$

$$\psi(0,y) = \psi \left(\frac{L}{H}, y\right) = 0 \quad . \tag{20}$$

The boundaries beneath the cap rock are impermeable to flow and thermally conductive; i.e.,

$$\psi\ (x,\eta) = 0 \quad , \tag{21}$$

$$\lambda_{cap} \frac{\partial\theta}{\partial y}\ (x,\eta) = \lambda_m \frac{\partial\theta}{\partial y}\ (x,\eta) \quad . \tag{22}$$

It is assumed that $\lambda_{cap} = \lambda_m$ so that Eq. (16) applies to both the cap rock and the permeable regions.

The boundaries at $y = 0$ are impermeable to flow and thermally nonconductive; i.e.,

$$\psi(x,0) = 0 \quad , \tag{23}$$

and

$$\frac{\partial\theta}{\partial y}\ (x,0) = 0 \quad . \tag{24}$$

NUMERICAL METHOD

The energy equation (16) is solved numerically by the Alternating Direction Implicit (ADI) method,[7] and the flow field equation (17) by the Gauss-Seidel iteration method. The region is divided into a uniform mesh, as shown in Fig. 2. The coordinates of the grid points are given by (x_i, y_j), where $x_i = (i-1)\Delta x$ and $y_j = (j-1)\Delta y$.



FIG. 2. Uniform mesh for the finite difference numerical solution.

A second-order finite-difference approximation formula is used for all spatial derivatives and a first-order finite-difference approximation for all time derivatives. The upwind scheme for the convection term is not used, but the numerical formulation can be easily adapted to the upwind scheme.

The ADI formulation of the energy equation (16) follows. First, the finite difference approximation for (2n+1)*th* time step is given as

$$\frac{\theta^{2n+1}_{i,j} - \theta^{2n}_{i,j}}{\Delta t} + u^{2n}_{i,j}\left(\frac{\theta^{2n+1}_{i+1,j} - \theta^{2n+1}_{i-1,j}}{2\Delta x}\right) + v^{2n}_{i,j}\left(\frac{\theta^{2n}_{i,j+1} - \theta^{2n}_{i,j-1}}{2\Delta y}\right)$$
$$= \frac{\theta^{2n+1}_{i+1,j} - 2\theta^{2n+1}_{i,j} + \theta^{2n+1}_{i-1,j}}{(\Delta x)^2} + \frac{\theta^{2n}_{i,j+1} - 2\theta^{2n}_{i,j} + \theta^{2n}_{i,j-1}}{(\Delta y)^2}, \quad (25)$$

where

$$u^{2n}_{i,j} = \gamma\left(\frac{\psi^{2n}_{i,j+1} - \psi^{2n}_{i,j-1}}{2\Delta y}\right), \quad (26)$$

$$v^{2n}_{i,j} = -\gamma\left(\frac{\psi^{2n}_{i+1,j} - \psi^{2n}_{i-1,j}}{2\Delta x}\right). \quad (27)$$

Equation (25) can be rewritten as

$$\left(-\frac{u^{2n}_{i,j}}{2\Delta x} - \frac{1}{(\Delta x)^2}\right)\theta^{2n+1}_{i-1,j} + \left(\frac{1}{\Delta t} + \frac{2}{(\Delta x)^2}\right)\theta^{2n+1}_{i,j}$$
$$+\left(\frac{u^{2n}_{i,j}}{2\Delta x} - \frac{1}{(\Delta x)^2}\right)\theta^{2n+1}_{i+1,j} = \frac{1}{\Delta t}\theta^{2n}_{i,j}$$
$$- v^{2n}_{i,j}\left(\frac{\theta^{2n}_{i,j+1} - \theta^{2n}_{i,j-1}}{2\Delta y}\right) + \frac{\theta^{2n}_{i,j+1} - 2\theta^{2n}_{i,j} + \theta^{2n}_{i,j-1}}{(\Delta y)^2}. \quad (28)$$

Equation (28) is valid for all grid points. At the boundary, both Eq. (28) and appropriate boundary conditions must be satisfied. We will now describe the finite difference equation for each boundary surface.

At the vertical boundary x = 0 (i.e., i = 1), the condition $\partial\theta/\partial x = 0$ requires that

$$\theta^k_{0,j} = \theta^k_{2,j} \quad . \tag{29}$$

Note that $\theta^k_{0,j}$ is a grid point outside of the region of interest at any time step k. With the aid of Eq. (29), Eq. (28) becomes

$$\left(\frac{1}{\Delta t} + \frac{2}{(\Delta x)^2}\right)\theta^{2n+1}_{1,j} - \frac{2}{(\Delta x)^2}\theta^{2n+1}_{2,j} = \frac{1}{\Delta t}\theta^{2n}_{1,j}$$

$$- v^{2n}_{1,j}\left(\frac{\theta^{2n}_{1,j+1} - \theta^{2n}_{1,j-1}}{2\Delta y}\right) + \frac{\theta^{2n}_{1,j+1} - 2\theta^{2n}_{1,j} + \theta^{2n}_{1,j-1}}{(\Delta y)^2} \tag{30}$$

for $2 \le j \le M - 1$.

At the vertical boundary x = L/H (i.e., i = N), the condition $\partial\theta/\partial x = 0$ requires that

$$\theta^k_{N+1,j} = \theta^k_{N-1,j} \quad . \tag{31}$$

Combining Eqs. (31) and (28), we obtain

$$- \frac{2}{(\Delta x)^2}\theta^{2n+1}_{N-1,j} + \left(\frac{1}{\Delta t} + \frac{2}{(\Delta x)^2}\right)\theta^{2n+1}_{N,j} = \frac{1}{\Delta t}\theta^{2n}_{N,j}$$

$$- v^{2n}_{N,j}\left(\frac{\theta^{2n}_{N,j+1} - \theta^{2n}_{N,j-1}}{2\Delta y}\right) + \left(\frac{\theta^{2n}_{N,j+1} - 2\theta^{2n}_{N,j} + \theta^{2n}_{N,j-1}}{(\Delta y)^2}\right) \tag{32}$$

for $2 \le j \le M - 1$.

At the lower boundary y = 0 (i.e., j = 1), the boundary condition $\partial\theta/\partial y = 0$ requires that

$$\theta^k_{i,0} = \theta^k_{i,2} \quad . \tag{33}$$

Combining Eqs. (33) and (28), we obtain

$$\left(- \frac{u_{i,1}^{2n}}{2\Delta x} - \frac{1}{(\Delta x)^2} \right) \theta_{i-1,1}^{2n+1} + \frac{1}{\Delta t} + \left(\frac{2}{(\Delta x)^2} \right) \theta_{i,1}^{2n+1} + \left(\frac{u_{i,1}^{2n}}{2\Delta x} - \frac{1}{(\Delta x)^2} \right) \theta_{i+1,1}^{2n+1} = \frac{1}{\Delta t} \theta_{i,1}^{2n} + \frac{2}{(\Delta y)^2} \left(\theta_{i,2}^{2n} - \theta_{i,1}^{2n} \right) \tag{34}$$

for $2 \leq i \leq N - 1$.

At $y = 1$, the boundary condition is

$$\theta_{i,M}^{k} = 0 \tag{35}$$

for $1 \leq i \leq N$.

Equations (28), (29), and (33) lead to

$$\left(\frac{1}{\Delta t} + \frac{2}{(\Delta x)^2} \right) \theta_{1,1}^{2n+1} - \frac{2}{(\Delta x)^2} \theta_{2,1}^{2n+1} = \frac{1}{\Delta t} \theta_{1,1}^{2n} + \frac{2}{(\Delta y)^2} \left(\theta_{1,2}^{2n} - \theta_{1,1}^{2n} \right) . \tag{36}$$

Equations (28), (31), and (33) lead to

$$- \frac{2}{(\Delta x)^2} \theta_{N-1,1}^{2n+1} + \left(\frac{1}{\Delta t} + \frac{2}{(\Delta x)^2} \right) \theta_{N,1}^{2n+1} = \frac{1}{\Delta t} \theta_{N,1}^{2n} + \frac{2}{(\Delta y)^2} \left(\theta_{N,2}^{2n} - \theta_{N,1}^{2n} \right) . \tag{37}$$

Equations (28), (30), (32), (34), (36), and (37) consist of M - 1 sets of N simultaneous equations of the form

$$B\theta_{1,j}^{2n+1} + C_{1,j}\theta_{2,j}^{2n+1} = D_{1,j} \tag{38}$$

for $1 \leq j \leq M - 1$,

$$A_{i,j}\theta^{2n+1}_{i-1,j} + B\theta^{2n+1}_{i,j} + C_{i,j}\theta^{2n+1}_{i+1,j} = D_{i,j} \tag{39}$$

for $2 \leq i \leq N - 1,\ 1 \leq j \leq M - 1$,

$$A_{N,j}\theta^{2n+1}_{N-1,j} + B\theta^{2n+1}_{N,j} = D_{N,j} \tag{40}$$

for $1 \leq j \leq M - 1$,

where

$$B = \frac{1}{\Delta t} + \frac{2}{(\Delta x)^2} \quad , \tag{41}$$

$$C_{1,j} = \frac{2}{(\Delta x)^2} \tag{42}$$

for $1 \leq j \leq M - 1$,

$$C_{i,j} = \frac{u^{2n}_{i,j}}{2\Delta x} - \frac{1}{(\Delta x)^2} \tag{43}$$

for $2 \leq i \leq N,\ 1 \leq j \leq M - 1$,

$$A_{i,j} = -\frac{u^{2n}_{i,j}}{2\Delta x} - \frac{1}{(\Delta x)^2} \tag{44}$$

for $1 \leq j \leq M - 1,\ 1 \leq i \leq N - 1$,

$$A_{N,j} = \frac{-2}{(\Delta x)^2} \tag{45}$$

for $1 \leq j \leq M - 1$,

$$D_{i,j} = \frac{1}{\Delta t}\,\theta^{2n}_{i,j} - v^{2n}_{i,j}\left(\frac{\theta^{2n}_{i,j+1} - \theta^{2n}_{i,j-1}}{2\Delta y}\right) + \frac{\theta^{2n}_{i,j+1} - 2\theta^{2n}_{i,j} + \theta^{2n}_{i,j-1}}{(\Delta y)^2} \tag{46}$$

for $2 \leq j \leq M - 1,\ 1 \leq i \leq N - 1$,

$$D_{i,1} = \frac{1}{\Delta t} \theta_{i,1}^{2n} + \frac{2}{(\Delta y)^2} \left(\theta_{i,2}^{2n} - \theta_{i,1}^{2n}\right) \tag{47}$$

for $1 \leq i \leq N - 1$.

The solution of Eqs. (38), (39), and (40) can be obtained in a straightforward manner.[7] Let

$$w_1 = B \quad , \tag{48}$$

$$w_i = B - A_{i,j}B_{i-1} \tag{49}$$

for $2 \leq i \leq N, 1 \leq j \leq M - 1$,

$$b_i = \frac{C_{i,j}}{w_i} \tag{50}$$

for $1 \leq j \leq M - 1, 1 \leq i \leq N - 1$,

$$g_1 = \frac{D_{1,j}}{w_1} \tag{51}$$

for $1 \leq j \leq M - 1$,

$$g_i = \frac{D_{i,j} - A_{i,j}g_{i-1}}{w_i} \tag{52}$$

for $2 \leq i \leq N - 1, 1 \leq j \leq M - 1$.

The solutions of the tridiagonal system are

$$\theta_{N,j}^{2n+1} = g_N \tag{53}$$

for $1 \leq j \leq M - 1$,

$$\theta_{i,j}^{2n+1} = g_i - b_i\theta_{i+1,j}^{2n+1} \tag{54}$$

for $1 \leq i \leq N - 1, 1 \leq j \leq M - 1$.

The computational procedure used to obtain solutions of the tridiagonal system for each set of the N simultaneous equations is the following. For a given j(jth set of equations where j is from 1 to M - 1), Eqs. (48) through (54) are computed with ascending value of i from 1 to N. After Eqs. (48) through (54) are evaluated, proceed to evaluate Eqs. (54) and (55) with decreasing value of i from N to 1. The values of the temperature function are stored in temporary storage location to allow evaluation of Eqs. (48) through (54) at previous time step temperature values.

The difference equation for Eq. (16) at (2n+2)*th* time step is given as

$$\frac{\theta_{i,j}^{2n+2} - \theta_{i,j}^{2n+1}}{\Delta t} + u_{i,j}^{2n+1}\left(\frac{\theta_{i+1,j}^{2n+1} - \theta_{i-1,j}^{2n+1}}{2\Delta x}\right) + v_{i,j}^{2n+1}\left(\frac{\theta_{i,j+1}^{2n+2} - \theta_{i,j-1}^{2n+2}}{2\Delta y}\right) = \frac{\theta_{i+1,j}^{2n+1} - 2\theta_{i,j}^{2n+1} + \theta_{i-1,j}^{2n+1}}{(\Delta x)^2} + \frac{\theta_{i,j+1}^{2n+2} - 2\theta_{i,j}^{2n} + \theta_{i,j-1}^{2n+2}}{(\Delta y)^2} \quad . \tag{55}$$

Equation (55) can be rewritten as

$$\left(-\frac{v_{i,j}^{2n+1}}{2\Delta y} - \frac{1}{(\Delta y)^2}\right)\theta_{i,j-1}^{2n+2} + \left(\frac{1}{\Delta t} + \frac{2}{(\Delta y)^2}\right)\theta_{i,j}^{2n+2} + \frac{v_{i,j}^{2n+1}}{2\Delta y} - \frac{1}{(\Delta y)^2} \quad \theta_{i,j+1}^{2n+2} = \frac{1}{\Delta t}\theta_{i,j}^{2n+1} - u_{i,j}^{2n+1}\left(\frac{\theta_{i+1,j}^{2n+1} - \theta_{i-1,j}^{2n+1}}{2\Delta x}\right) + \frac{\theta_{i+1,j}^{2n+1} - 2\theta_{i,j}^{2n+1} + \theta_{i-1,j}^{2n+1}}{(\Delta x)^2} \tag{56}$$

for $2 \leq i \leq N - 1,\ 2 \leq j \leq M - 1$.

Equation (56), when combined with boundary conditions (29), (31), (33), and (35), results in the following equations:

$$\left(-\frac{v_{1,j}^{2n+1}}{2\Delta y}-\frac{1}{(\Delta y)^2}\right)\theta_{1,j-1}^{2n+2}+\frac{1}{\Delta t}+\frac{2}{(\Delta y)^2}\quad\theta_{1,j}^{2n+2}$$
$$+\left(\frac{v_{1,j}^{2n+1}}{2\Delta y}-\frac{1}{(\Delta y)^2}\right)\theta_{1,j+1}^{2n+2}=\frac{1}{\Delta t}\theta_{1,j}^{2n+1}+\frac{2}{(\Delta x)^2}\left(\theta_{2,j}^{2n+1}-\theta_{1,j}^{2n+1}\right) \tag{57}$$

for $2 \leq j \leq M - 1$,

$$\left(-\frac{v_{N,j}^{2n+1}}{2\Delta y}-\frac{1}{(\Delta y)^2}\right)\theta_{N,j-1}^{2n+2}+\left(\frac{1}{\Delta t}+\frac{2}{(\Delta y)^2}\right)\theta_{N,j}^{2n+2}$$
$$+\left(\frac{v_{N,j}^{2n+1}}{2\Delta y}-\frac{1}{(\Delta y)^2}\right)\theta_{N,j+1}^{2n+2}=\frac{1}{\Delta t}\theta_{N,j}^{2n+1}+\frac{2}{(\Delta x)^2}\left(\theta_{N-1,j}^{2n+1}-\theta_{N,j}^{2n+1}\right) \tag{58}$$

for $2 \leq j \leq M - 1$,

$$\left(\frac{1}{\Delta t}+\frac{2}{(\Delta y)^2}\right)\theta_{i,1}^{2n+2}-\frac{2}{(\Delta y)^2}\theta_{i,2}^{2n+2}=\frac{1}{\Delta t}\theta_{i,1}^{2n+1}$$
$$-u_{i,1}^{2n+1}\left(\frac{\theta_{i+1,1}^{2n+1}-\theta_{i-1,1}^{2n+1}}{2\Delta x}\right)+\frac{\theta_{i+1,1}^{2n+1}-2\theta_{i,1}^{2n+1}\,\theta_{i-1,1}^{2n+1}}{(\Delta x)^2} \tag{59}$$

for $2 \leq i \leq N - 1$,

$$\left(\frac{1}{\Delta t}+\frac{2}{(\Delta y)^2}\right)\theta_{1,1}^{2n+2}-\frac{2}{(\Delta y)^2}\theta_{1,2}^{2n+2}=\frac{1}{\Delta t}\theta_{1,1}^{2n+1}$$
$$+\frac{2}{(\Delta x)^2}\left(\theta_{2,1}^{2n+1}-\theta_{1,1}^{2n+1}\right)\ , \tag{60}$$

$$\left(\frac{1}{\Delta t}+\frac{2}{(\Delta y)^2}\right)\theta_{N,1}^{2n+2}-\frac{2}{(\Delta y)^2}\theta_{N,2}^{2n+2}=\frac{1}{\Delta t}\theta_{N,1}^{2n+1}$$
$$+\frac{2}{(\Delta x)^2}\left(\theta_{N-1,1}^{2n+1}-\theta_{N,1}^{2n+1}\right)\ . \tag{61}$$

Equations (56) through (61) consist of N sets of (M - 1) simultaneous equations of the form

$$B\theta_{i,1}^{2n+2}+C_{i,1}\theta_{i,2}^{2n+2}=D_{i,1} \tag{62}$$

for $1 \leq i \leq N$,

$$A_{i,j}\theta^{2n+2}_{i,j-1} + B\theta^{2n+2}_{i,j} + C_{i,j}\theta^{2n+2}_{i,j+1} = D_{i,j} \tag{63}$$

for $1 \leq i \leq N,\ 2 \leq j \leq M - 2$,

$$A_{i,M-1}\theta^{2n+2}_{i,M-2} + B\theta^{2n+2}_{i,M-1} = D_{i,M-1} \tag{64}$$

for $1 \leq i \leq N$,

where

$$B = \frac{1}{\Delta t} + \frac{2}{(\Delta y)^2} , \tag{65}$$

$$C_{i,1} = \frac{-2}{(\Delta y)^2} \tag{66}$$

for $1 \leq i \leq N$,

$$C_{i,j} = \frac{V^{2n+1}_{i,j}}{2\Delta y} - \frac{1}{(\Delta y)^2} \tag{67}$$

for $1 \leq i \leq N,\ 2 \leq j \leq M - 1$,

$$A_{i,j} = -\frac{V^{2n+1}_{i,j}}{2\Delta y} - \frac{1}{(\Delta y)^2} \tag{68}$$

for $1 \leq i \leq N,\ 2 \leq j \leq M - 1$,

$$D_{i,j} = \frac{1}{\Delta t}\,\theta^{2n+1}_{i,j} - u^{2n+1}_{i,j}\left(\frac{\theta^{2n+1}_{i+1,j} - \theta^{2n+1}_{i-1,j}}{2\Delta x}\right) + \frac{\theta^{2n+1}_{i+1,j} - 2\theta^{2n+1}_{i,j} + \theta^{2n+1}_{i-1,j}}{(\Delta x)^2} \tag{69}$$

for $2 \leq i \leq N - 1,\ 1 \leq j \leq M - 1$,

$$D_{1,j} = \frac{1}{\Delta t}\,\theta^{2n+1}_{1,j} + \frac{2}{(\Delta x)^2}\left(\theta^{2n+1}_{2,j} - \theta^{2n+1}_{1,j}\right) \tag{70}$$

for $1 \leq j \leq M - 1$,

$$D_{N,j} = \frac{1}{\Delta t} \theta^{2n+1}_{N,j} + \frac{2}{(\Delta x)^2} \left(\theta^{2n+1}_{N-1,j} - \theta^{2n+1}_{N,j} \right) \tag{71}$$

for $1 \leq j \leq M - 1$.

The solutions of the N sets of tridiagonal systems can be obtained in a straightforward manner. Let

$$w_1 = B \quad , \tag{72}$$

$$b_j = \frac{C_{i,j}}{w_j} \tag{73}$$

for $1 \leq i \leq N$, $1 \leq j \leq M - 2$,

$$w_j = B - A_{i,j} b_{j-1} \tag{74}$$

for $1 \leq i \leq N$, $2 \leq j \leq M - 1$,

$$g_1 = \frac{D_{i,1}}{w_i} \quad , \tag{75}$$

$$g_j = \frac{D_{i,j} - A_{i,j} g_{j-1}}{w_j} \tag{76}$$

for $1 \leq i \leq N$, $2 \leq j \leq M - 1$.

The solutions are

$$\theta^{2n+2}_{i,M-1} = g_{M-1} \tag{77}$$

for $1 \leq i \leq N$,

$$\theta^{2n+2}_{i,j} = g_j - b_j \theta^{2n+2}_{i,j+1} \tag{78}$$

for $1 \leq j \leq M - 2$, $1 \leq i \leq N$.

A second-order finite-difference approximation for the stream function equation (17) is

$$\frac{\psi_{i+1,j} - 2\psi_{i,j} + \psi_{i-1,j}}{(\Delta x)^2} + \chi \frac{\psi_{i,j+1} - 2\psi_{i,j} + \psi_{i,j-1}}{(\Delta y)}$$

$$= -Ra \left(\frac{\theta^{2n+1}_{i+1,j} - \theta^{2n+1}_{i-1,j}}{2\Delta x} \right) \quad (79)$$

for $1 \leq i \leq N - 1,\ 1 \leq j \leq M - 1$.

Equation (79) can be rewritten as

$$\psi_{i,j} = \frac{1}{2(1 + \varepsilon)} \left[\psi_{i+1,j} + \psi_{i-1,j} + \varepsilon\psi_{i,j+1} + \varepsilon\psi_{i,j-1} + \frac{Ra\ (\Delta x)}{2} \left(\theta^{2n+1}_{i+1,j} - \theta^{2n+1}_{i-1,j}\right)\right] \quad (80)$$

for $1 \leq i \leq N - 1,\ 1 \leq j \leq M - 1$,

where

$$\varepsilon = \chi\ (\Delta y/\Delta x)^2 \ . \quad (81)$$

NUMERICAL RESULTS

The flow field (stream function, velocity) is initialized to zero everywhere in the flow region. The temperature field is zero everywhere except in the region of an intrusive dike, where it is equal to 1. Figure 3 charts the numerical computation procedure, which is as follows:

1. Initial data values are set to conform with initial conditions of the problem.

2. Temperature field solutions are obtained for (2n+1)*th* time step using Eqs. (53) and (54).

3. The stream function equation (80) is solved by the Gauss-Seidel iteration method. The iteration is terminated when maximum change in stream function values is less than 10^{-5} during two successive iteration cycles.

4. Velocity components are computed using Eqs. (26) and (27).

5. Temperature field solutions are obtained for (2n+2)*th* time step using Eqs. (77) and (78).

6. Steps 3 and 4 are performed again.

7. If desired, the temperature, stream function, velocity vector, and surface heat flow can be plotted.

8. If the maximum time step is reached, then the program is terminated. Otherwise a return to step 2 is required.



FIG. 3. Flowchart diagram of the numerical computation procedures.

The reservoir parameter values used in the numerical computation are

Parameter	Value
K_x (permeability), mD	160
H (depth), m	6,000
L (width), m	12,000
λ_m (conductivity), W/(m•K)	3.3
α_m (diffusivity), m^2/s	1.33×10^{-6}
ΔT (maximum temperature), K	700

The surface heat flow in terms of the dimensionless thermal gradient $\partial\theta/\partial y$ is given by

$$Q = \lambda_m \frac{\partial\theta'}{\partial y'} = \lambda_m \frac{\Delta T}{H} \frac{\partial\theta}{\partial y} = 8.6 \frac{\partial\theta}{\partial y} \text{ (HFU)} \quad . \tag{82}$$

The relationship between real time (t') and dimensionless time (t) is given by

$$t' = \frac{H^2}{\alpha_m} t = 870{,}000 \ t \text{ (in years)} \quad . \tag{83}$$

Figures 4 through 6 show the graphs of temperature, stream function, velocity vector, and surface heat flow produced by the cooling of an intrusive dike complex 1500 m in width and 3900 m in height located at the left boundary. All results were obtained with Ra = 200 and time (t') = 10,400 y. Figure 4 is obtained with $\chi = 2$, Fig. 5 with $\chi = 0.25$, Fig. 6 with $\chi = 0.5$.

It is interesting to note from Figs. 4 and 5 that the surface heat flow is higher for the case of lower permeability ratio (χ). One can explain this by observing the flow patterns in these figures. For the case of the higher χ, the flow is behaving like the flow near a vertical flat plate and therefore produces very little convection of heat from the top of the dike region to the surface. On the other hand the lower permeability ratio (χ) produces large convective flow on the top of the dike region. Figures 6 through 8 present the effects of the dike's vertical dimension on surface heat flow. It is quite clear that the closer the top of the intrusion is to the surface, the higher the resulting surface heat flow.

Figure 9 presents the history of surface heat flow. The sharp drop-off of surface heat flow in the Salton Sea Geothermal Field (SSGF) as noted by Kasameyer and Younker[6] is consistent with these numerical results. Figure 10 presents the temperature contour plots at various time steps. A simple analytic model by Hanson[8] involving horizontal convection transport beneath a conductive cap suggests that the age of the intrusive body is between 6000 and 20,000 y, based on field data from the SSGF. Figure 9 provides more data substantiating this estimate of the age of the intrusive dike. In Fig. 11, the results indicate that when χ is very small, multilayer convective cells exist.

The appendix contains the finite-difference heat and mass transport computer program used for the above calculations.



FIG. 4. Temperature (a), stream function (b), velocity (c), and surface heat flow (d) produced by cooling of intrusive dike located at left boundary. Ra = 200, χ = 2.0, η = 0.9, and t = 0.012.



FIG. 5. Temperature (a), stream function (b), velocity (c), and surface heat flow (d) produced by cooling of intrusive dike located at left boundary. Ra = 200, $\chi = 0.25$, $\eta = 0.9$, and $t = 0.012$.



FIG. 6. Effect of the vertical dimension on temperature (a), stream function (b), velocity (c), and surface heat flow (d) during cooling of intrusive dike located at left boundary. Ra = 200, χ = 0.5, η = 0.9, and t = 0.012.



FIG. 7. Effect of the vertical dimension on temperature (a), stream function (b), velocity (c), and surface heat flow (d) during cooling of intrusive dike located at left boundary. Ra = 200, χ = 0.5, η = 0.9, and t = 0.012. Note change in size of dike.



FIG. 8. Effect of the vertical dimension on temperature (a), stream function (b), velocity (c), and surface heat flow (d) during cooling of intrusive dike located at left boundary. Ra = 200, $\chi = 0.5$, $\eta = 0.9$, and t = 0.012. Note change in size of dike.



FIG. 9. History of surface heat flow for dike located at left boundary. Ra = 200, $\chi = 0.5$, $\eta = 0.9$ with t = 0.004 at (a), t = 0.008 at (b), t = 0.012 at (c), t = 0.016 at (d), and t = 0.020 at (e).



FIG. 10. Temperature contour plots for dike located at left boundary. Ra = 200, χ = 0.5, η = 0.9 with t = 0.004 at (a), t = 0.012 at (b), and t = 0.02 at (c).



FIG. 11. Multilayer convective cells at low χ ratio.

ACKNOWLEDGMENTS

This work was supported by Lawrence Livermore National Laboratory (LLNL) Summer Institute Program. Invaluable suggestions were received from LLNL scientists J. M. Hanson, who proposed this research problem; P. W. Kasameyer; and L. W. Younker, formerly of the Department of Geological Sciences, University of Illinois, Chicago.

REFERENCES

1. J. C. Jaeger, "Thermal Effects of Intrusions," Rev. Geophys. 2, 443-446 (1964).
2. K. Horai, "Heat Flow Anomaly Associated with Dike Intrusion," J. Geophys. Rev. 79, 1940-1946 (1974).
3. K. H. Lau and P. Cheng, "The Effect of Dike Intrusion on Free Convection in Conduction-dominated Geothermal Reservoirs," Int. J. Heat Mass Transfer 20, 1205-1210 (1977).
4. D. Norton and J. Knight, "Transport Phenomena in Hydrothermal Systems: Cooling Plutons," Am. J. Sci. 277, 937-981 (1977).
5. K. E. Torrance and J. P. Sheu, "Heat Transfer from Plutons Undergoing Hydrothermal Cooling and Thermal Cracking," Numerical Heat Transfer 1, 147-161 (1978).
6. P. W. Kasameyer and L. W. Younker, "Natural Fluid-flow Patterns in the Salton Sea Geothermal Field," Geothermal Resources Council, Transactions 2, 359-361 (1978).
7. D. W. Peaceman and H. H. Rachford, Jr., "The Numerical Solution of Parabolic and Elliptic Differential Equations," J. Soc. Indust. Appl. Math. 3, No. 1, 28-41 (1955).
8. J. M. Hanson, Lawrence Livermore National Laboratory, Livermore, California, private communication (1979).

CG/lf

APPENDIX:
COMPUTER PROGRAM

```
*******         CHAT  170A   BOX W42  07:51:30 08/08/79R

C        ---------------------------------------------------------------
C        VARIABLES DESCRIPTIONS
C        ---------------------------------------------------------------
C        T(I,J)     TEMPERATURE VALUE AT GRID POINT (I,J)
C        S(I,J)     STREAM FUNCTION VALUE AT GRID POINT (I,J)
C        U(I,J)     VELOCITY COMPONENT IN X-DIRECTION AT GRID POINT (I,J)
C        V(I,J)     VELOCITY COMPONENT IN Y-DIRECTION AT GRID POINT (I,J)
C        KRATIO     VERTICAL AND HORIZONTAL PERMEABILITY RATIO
C        IPWS       LEFT BOUNDARY OF DIKE
C        IPW        RIGHT BOUNDARY OF DIKE
C        IPHI       HEIGHT OF THE DIKE
C        IPH        HEIGHT OF THE DIKE
C        IUH        LOCATION OF THE CAP ROCK
C        IMAX       MAXIMUM NUMBER OF POINT IN X-DIRECTION
C        JMAX       MAXIMUM NUMBER OF POINT IN Y-DIRECTION
C        DELT       INCREMENTAL VALUE OF EACH TIME STEP
C        DELX       INCREMENTAL VALUE OF EACH GRID POINT IN X-DIRECTION
C        DELY       INCREMENTAL VALUE OF EACH GRID POINT IN Y-DIRECTION
C        RELAX      RELAXATION FACTOR USED IN STREAM FUNCTION ITERATION
C        R          RAYLEIGH NUMBER
C        CYMAX      MAXIMUM NUMBER OF TIME STEPS DESIRED
C        IPLOT      NUMBER OF TIME STEPS BETWEEN TWO RJET PLOTS
C        ---------------------------------------------------------------

      PROGRAM GEOTHERMAL(TAPE59,TAPE61)
      REAL KRATIO
      DIMENSION T(61,21),S(61,21),U(61,21),V(61,21)
      DIMENSION CL(6)
      DIMENSION CS(6)
      DIMENSION X(61),Y(61),W(61),G(61),B(61),TS(61)
      DATA T/1281*0./
      DATA S/1281*0./
      DATA U/1281*0./
      DATA V/1281*0./

C       ----------------------------------------------------------------
C         STATEMENT FUNCTION USED BY ADI SOLUTION
C       ----------------------------------------------------------------

      D(I,J)=DELTIN*T(I,J)-V(I,J)*(T(I,J+1)-T(I,J-1))/(2.*DELY)+
     1   DELY2I*(T(I,J+1)-2.*T(I,J)+T(I,J-1))
      D1(I,J)=DELTIN*T(I,J)-U(I,J)*(T(I+1,J)-T(I-1,J))/(2.*DELX) +
     1   DELX2I*(T(I+1,J)-2.*T(I,J)+T(I-1,J))

C       ----------------------------------------------------------------
C         PROGRAM STARTS HERE
C       ----------------------------------------------------------------

      CALL CHANGE("+GEOTH1")
      CALL ASSIGN(61,6HPRINT1)
      CALL RJETID


C       ----------------------------------------------------------------
C         TEMPERATURE FIELD PLOTTING LEVEL VALUES
C       ----------------------------------------------------------------
*******
```

```
      CL(1)=0.1
      DO 50 I=2,6
      CL(I)=CL(I-1)+0.2
50    CONTINUE

C     ---------------------------------------------------------------------
C       PARAMETERS OF THE PROBLEM
C     ---------------------------------------------------------------------

      IPWS=1
      IPW=6
      CYMAX=20
      IPLOT=4
      IUH=20
      IPHI=14
      IPH=IPHI
      DELT=0.001
      IMAX=41
      JMAX=21
      R=200.
      KRATIO=0.01

C     ---------------------------------------------------------------------
C       OTHER COMPUTATIONAL CONSTANTS
C     ---------------------------------------------------------------------

      IMAX1=IMAX-1
      JMAX1=JMAX-1
      IMAX2=IMAX-2
      JMAX2=JMAX-2
      TIME=0.
      ICYC=0
      DELY=1.0/FLOAT(JMAX1)
      DELX=DELY
      XMAX=IMAX1*DELX
      DELTIN=1./DELT
      DELX2I=1./(DELX*DELX)
      DELY2I=1./(DELY*DELY)
      IPH1=IPH+1
      RELAX=0.8
      IUH1=IUH-1
      EPSI=KRATIO*DELX*DELY2I*DELX
      RX=R*DELX*DELX
      SMAX=0.
      SDEL=0.
      X(1)=0.
      Y(1)=0.
      DO 11 I=2,IMAX
      X(I)=X(I-1)+DELX
11    CONTINUE
      DO 12 I=2,JMAX
      Y(I)=Y(I-1)+DELY
12    CONTINUE

C     ---------------------------------------------------------------------
C       SET INITIAL TEMPERATURE FIELD VALUES
```

```
      C     ------------------------------------------------------------

            DO 10 I=IPWS,IPW
            DO 10 J=1,IPHI
            T(I,J)=1.0
       10   CONTINUE

      C     ------------------------------------------------------------
      C        ITERATION LOOP STARTS HERE
      C     ------------------------------------------------------------

      1000  CONTINUE

      C     ------------------------------------------------------------
      C     START ADI ITERATION FOR TEMPERATURE FIELD
      C      ADI IN X-DIRECTION
      C        FOR Y=0
      C     ------------------------------------------------------------

            ITIME=1
            W(1)=DELTIN + 2.*DELX2I
            B(1)= -2.*DELX2I/W(1)
            G(1)= DELTIN * T(1,1) + 2.*DELY2I*(T(1,2)-T(1,1))
            G(1)=G(1)/W(1)
            DO 100 I=2,IMAX
            A=U(I,1)/(2.*DELX) + DELX2I
            W(I)=W(1) + A*B(I-1)
            B(I)=(U(I,1)/(2.*DELX)-DELX2I)/W(I)
            G1=DELTIN*T(I,1) + 2.*DELX2I*(T(I,2)-T(I,1))
            G(I)=(G1+A*G(I-1))/W(I)
       100  CONTINUE

      C     ------------------------------------------------------------
      C        SOLUTION FOR TEMPERATURE FIELD AT Y=0
      C        STORE SOLUTION IN TEMPORARY STORAGE
      C     ------------------------------------------------------------

            TS(IMAX)=G(IMAX)
            DO 101 I=1,IMAX1
            I1=IMAX-I
            TS(I1)=G(I1)-B(I1)*TS(I1+1)
       101  CONTINUE


      C     ------------------------------------------------------------
      C        SOLUTION FOR TEMPERATURE FIELD AT ALL OTHER Y
      C        IMPLICIT SOLUTION FOR ALL OTHER Y IN X-DIRECTION
      C     ------------------------------------------------------------

            DO 105 J=2,JMAX1
            G(1)=D(1,J)/W(1)
            DO 106 I=2,IMAX
            A=U(I,J)/(2.*DELX) +DELX2I
            W(I)=W(1) + A*B(I-1)
            B(I)=(U(I,J)/(2.*DELX)-DELX2I)/W(I)
            G(I)=(D(I,J)+A*G(I-1))/W(I)
       106  CONTINUE
```
****

```
C     ----------------------------------------------------------------
C       STORE SOLUTION FROM TEMPORARY STORAGE IN TEMPERATURE FIELD
C     ----------------------------------------------------------------

      DO 107 I=1,IMAX
      T(I,J-1)=TS(I)
107   CONTINUE

C     ----------------------------------------------------------------
C       SOLUTION IS STORED IN TEMPORARY STORAGE
C     ----------------------------------------------------------------

      TS(IMAX)=G(IMAX)
      DO 108 I=1,IMAX1
      I1=IMAX-I
      TS(I1)=G(I1)-B(I1)*TS(I1+1)
108   CONTINUE
105   CONTINUE

C     ----------------------------------------------------------------
C       STORE SOLUTION FROM TEMPORARY STORAGE
C          INTO TEMPERATURE FIELD
C     ----------------------------------------------------------------

      DO 109 I=1,IMAX
      T(I,JMAX1)=TS(I)
109   CONTINUE
      TIME=TIME+DELT
      ICYC=ICYC+1
      GO TO 305


C     ----------------------------------------------------------------
C       SECOND HALF OF ADI SOLUTION STARTS HERE
C     ----------------------------------------------------------------


C     ----------------------------------------------------------------
C       SOLUTION FOR X=0
C     ----------------------------------------------------------------

180   CONTINUE
      ITIME=2
      W(1)=DELTIN + 2.*DELY2I
      B(1)= -2.*DELY2I/W(1)
      G(1)=DELTIN*T(1,1) + 2.*DELX2I*(T(2,1)-T(1,1))
      G(1)=G(1)/W(1)
      DO 200 J=2,JMAX1
        A=V(1,J)/(2.*DELY) + DELY2I
      W(J)=W(1) + A*B(J-1)
      B(J)=(V(1,J)/(2.*DELY)-DELY2I)/W(J)
      G1=DELTIN*T(1,J) + 2.*DELX2I*(T(2,J)-T(1,J))
      G(J)=(G1 + A*G(J-1))/W(J)
200   CONTINUE

C     ----------------------------------------------------------------
*******
```

```
C        STORE SOLUTION IN TS TEMPORARY
C     -----------------------------------------------------------------------

      TS(JMAX1)=G(JMAX1)
      DO 201 J=1,JMAX2
      J1=JMAX1-J
      TS(J1)=G(J1)-B(J1)*TS(J1+1)
201   CONTINUE

C     -----------------------------------------------------------------------
C        SOLUTION FOR 0<X<L
C     -----------------------------------------------------------------------

      DO 205 I=2,IMAX1
      G(1)=D1(I,1)/W(1)
      DO 206 J=2,JMAX1
      A=V(I,J)/(2.*DELY) + DELY2I
      W(J)=W(1) + A*B(J-1)
      B(J)=(V(I,J)/(2.*DELY)-DELY2I)/W(J)
      G(J)=(D1(I,J) + A*G(J-1))/W(J)
206   CONTINUE

C     -----------------------------------------------------------------------
C        STORE SOLUTION INTO T
C     -----------------------------------------------------------------------

      DO 207 J=1,JMAX1
      T(I-1,J)=TS(J)
207   CONTINUE

C     -----------------------------------------------------------------------
C        STORE SOLUTION TEMPORARY IN TS
C     -----------------------------------------------------------------------

      TS(JMAX1)=G(JMAX1)
      DO 208 J=1,JMAX2
      J1=JMAX1-J
      TS(J1)=G(J1)-B(J1)*TS(J1+1)
208   CONTINUE
205   CONTINUE

C     -----------------------------------------------------------------------
C        SOLUTION FOR X=L
C     -----------------------------------------------------------------------

      G(1)=DELTIN*T(IMAX,1)+2.*DELX2I*(T(IMAX1,1)-T(IMAX,1))
      G(1)=G(1)/W(1)
      DO 210 J=2,JMAX1
      A=V(IMAX,J)/(2.*DELY) + DELY2I
      W(J)=W(1) + A*B(J-1)
      B(J)=(V(IMAX,J)/(2.*DELY) - DELY2I)/W(J)
      G1=DELTIN*T(IMAX,J) + 2.*DELX2I*(T(IMAX1,J)-T(IMAX,J))
      G(J)=(G1+A*G(J-1))/W(J)
210   CONTINUE

C        STORE SOLUTION INTO T

*******
```

```
      DO 211 J=1,JMAX1
      T(IMAX1,J)=TS(J)
211   CONTINUE
      TS(JMAX1)=G(JMAX1)
      DO 212 J=1,JMAX2
      J1=JMAX1-J
      TS(J1)=G(J1)-B(J1)*TS(J1+1)
212   CONTINUE
C     ---------------------------------------------------------------
C       SOLUTION OBTAINED FOR X=L
C     ---------------------------------------------------------------
      DO 213 J=1,JMAX1
      T(IMAX,J)=TS(J)
213   CONTINUE
      TIME=TIME+DELT
      ICYC=ICYC+1
C     ---------------------------------------------------------------
C       ENTER STREAM FUNCTION ITERATION LOOP
C     ---------------------------------------------------------------
305   DELS=0.
      SMAX=0.
      SMIN=0.
      DO 310 I=2,IMAX1
      IF(I.LT.IPWS)GO TO 311
      IF(I.GT.IPW)GO TO 311
      JSTART=IPH1
      GO TO 312
311   JSTART=2
312   DO 315 J=JSTART,IUH1
      ST=S(I+1,J)+S(I-1,J)+EPSI*(S(I,J+1)+S(I,J-1))
      ST=ST+RX*(T(I+1,J)-T(I-1,J))/(2.*DELX)
      ST=ST/(2*(1.+EPSI))
      ST1=S(I,J)
      S(I,J)=RELAX*ST + (1.-RELAX)*S(I,J)
      IF(SMAX.GT.S(I,J))GO TO 316
      SMAX=S(I,J)
316   CONTINUE
      IF(SMIN.LT.S(I,J))GO TO 317
      SMIN=S(I,J)
317   CONTINUE
      DELS1=ABS(ST1-S(I,J))
      IF(DELS1.LT.DELS)GO TO 315
      DELS=DELS1
315   CONTINUE
310   CONTINUE
C     ---------------------------------------------------------------
C       CHECK TO SEE IF ITERATION CONVERGENT CRITERIA WERE MET
C     ---------------------------------------------------------------
      IF(DELS.GT.1.0E-5)GO TO 305
```

****

```
C      ------------------------------------------------------------
C         PLOT ON RJET
C         PLOT TEMPERATURE
C      ------------------------------------------------------------

       IF(MOD(ICYC,IPLOT).NE.0)GO TO 4000
       CALL MAPS(0.,XMAX,0.,1.,0.11,1.0,0.11,0.43)
       CALL SETLCH(0.5,1.5,0,0,2,0)
       WRITE(100,3)TIME
3      FORMAT("TEMPERATURE   AT TIME = ",F7.3)
       CALL RCONTR(6,CL,0,T,61,X,1,IMAX,1,Y,1,JMAX,1)
       CALL FRAME

C      ------------------------------------------------------------
C         PLOT STREAM FUNCTION
C      ------------------------------------------------------------

       CS(1)=SMIN
       CS(6)=0.
       DS=(SMAX-SMIN)/5.0
       DO 360 I=2,5
       CS(I)=CS(I-1)+DS
360    CONTINUE
       CALL MAPS(0.,XMAX,0.,1.,0.11,1.0,0.11,0.43)
       CALL SETLCH(0.5,1.5,0,0,2,0)
       WRITE(100,4)
4      FORMAT("STREAM FUNCTION")
       CALL RCONTR(6,CS,0,S,61,X,1,IMAX,1,Y,1,JMAX,1)
       CALL FRAME

C      ------------------------------------------------------------
C         COMPUTE VELOCITY
C      ------------------------------------------------------------

4000   CONTINUE
       SMAX=0.
       DO 400 I=1,IMAX
       DO 400 J=1,IUH
       U(I,J)=(S(I,J+1)-S(I,J-1))/(2.*DELY)
       V(I,J)=(S(I-1,J)-S(I+1,J))/(2.*DELX)
       IF(ABS(U(I,J)).LE.SMAX)GO TO 410
       SMAX=ABS(U(I,J))
410    IF(ABS(V(I,J)).LE.SMAX)GO TO 400
       SMAX=ABS(V(I,J))
400    CONTINUE

C      ------------------------------------------------------------
C         PLOT VELOCITY
C      ------------------------------------------------------------

       IF(MOD(ICYC,IPLOT).NE.0)GO TO 4010
       CALL MAPS(0.,XMAX,0.,1.,0.11,1.0,0.11,0.43)
       CALL SETLCH(0.5,1.5,0,0,2,0)
       WRITE(100,5)
5      FORMAT("VELOCITY")
       DO 430 I=2,IMAX1
```
****

```
          DO 430 J=2,JMAX1
          XS=X(I)
          YS=Y(J)
          FAC=2.5
          XE=XS+FAC*U(I,J)*DELX/SMAX
          YE=YS+FAC*V(I,J)*DELY/SMAX
          CALL PLOTV(XS,YS,XE,YE)
  430     CONTINUE
          CALL FRAME

C          ------------------------------------------------------------------
C             PLOT TEMPERATURE PROFILES
C          ------------------------------------------------------------------

          CALL MAPS(0.,XMAX,0.,1.,0.11,1.0,0.11,0.43)
          CALL SETLCH(0.5,1.5,0,0,2,0)
          WRITE(100,6)
  6       FORMAT("TEMPERATURE PROFILES")
          DO 450 I=1,IMAX1,10
          IF(I.EQ.1)GO TO 451
          XS=X(I)
          CALL LINE(XS,1.0,XS,0.,0)
  451     DO 460 J=1,JMAX
          TS(J)=T(I,J)*0.5+X(I)
  460     CONTINUE
          CALL TRACE(TS,Y,JMAX)
  450     CONTINUE
          CALL FRAME

C          ------------------------------------------------------------------
C             PLOT SURFACE HEAT FLOW
C          ------------------------------------------------------------------

          HMAX=0.
          CALL MAPS(0.,XMAX,0.,20.,0.11,1.0,0.11,0.3)
          CALL SETLCH(0.5,24.,0,0,2,0)
          WRITE(100,7)
  7       FORMAT("SURFACE HEAT FLOW")
          DO 500 I=1,IMAX
          W(I)=(T(I,JMAX2)-T(I,JMAX))/(2.*DELY)
          W(I)=W(I)*8.6
  500     CONTINUE
          CALL TRACE(X,W,IMAX)

C          ------------------------------------------------------------------
C             PLOT TEMPERATURE BENEATH THE CAP
C          ------------------------------------------------------------------

          CALL MAPS(0.,XMAX,0.,0.5,0.11,1.0,0.61,0.8)
          DO 510 I=1,IMAX
          W(I)=T(I,IUH)
  510     CONTINUE
          CALL SETLCH(0.5,1.0,0,0,2,0)
          WRITE(100,8)
  8       FORMAT("TEMPERATURE BENEATH THE CAP")
          CALL TRACE(X,W,IMAX)
          CALL FRAME
```
****

```
*******          CHAT  170A   BOX W42  07:51:30 08/08/79R          MAIN.

          C      -------------------------------------------------------------------
          C         LOOP BACK FOR NEXT TIME STEP
          C      -------------------------------------------------------------------

          4010  CONTINUE
                IF(ITIME.EQ.1)GO TO 180
                IF(ICYC.LE.CYMAX)GO TO 1000

          C      -------------------------------------------------------------------
          C         PROGRAM END. WILL PRINT THE LAST TIME STEP TEMPERATURE AND
          C            STREAM FUNCTION VALUES.
          C      -------------------------------------------------------------------

                PRINT 9000
           9000 FORMAT("1   THIS IS THE TEMPERATURE DATA")
                DO 20 I=1,IMAX
                PRINT 9001,(T(I,J),J=1,JMAX)
          9001  FORMAT(1H ,11F10.5)
           20   CONTINUE
                PRINT 9002
          9002  FORMAT(1H1,"STREAM FUNCTION")
                DO 21 I=1,IMAX
                PRINT 9001,(S(I,J),J=1,JMAX)
          21    CONTINUE
                CALL EXIT
                END
*******
```

U. S. Government Printing Office: 1981/1—789-002/5540